Rules for integrands of the form $(d + e\,x)^m\,(f + g\,x)\,\left(a + b\,x + c\,x^2\right)^p$ when $e\,f - d\,g \neq 0$

0: $\int (e\,x)^m\,(f + g\,x)\,\left(b\,x + c\,x^2\right)^p\,dx$ when $b\,g\,(m + p + 1) - c\,f\,(m + 2\,p + 2) == 0 \land m + 2\,p + 2 \neq 0$

Rule 1.2.1.3.0: If $b\,g\,(m + p + 1) - c\,f\,(m + 2\,p + 2) == 0 \land m + 2\,p + 2 \neq 0$, then

$$\int (e\,x)^m\,(f + g\,x)\,\left(b\,x + c\,x^2\right)^p\,dx \longrightarrow \frac{g\,(e\,x)^m\,\left(b\,x + c\,x^2\right)^{p+1}}{c\,(m + 2\,p + 2)}$$

Program code:

```
Int[(e_.*x_)^m_.*(f_+g_.*x_)*(b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  g*(e*x)^m*(b*x+c*x^2)^(p+1)/(c*(m+2*p+2)) /;
FreeQ[{b,c,e,f,g,m,p},x] && EqQ[b*g*(m+p+1)-c*f*(m+2*p+2),0] && NeQ[m+2*p+2,0]
```

1: $\int x^m\,(f + g\,x)\,\left(a + c\,x^2\right)^p\,dx$ when $m \in \mathbb{Z} \land 2\,p \notin \mathbb{Z}$

Derivation: Algebraic expansion

Rule 1.2.1.3.1: If $m \in \mathbb{Z} \land 2\,p \notin \mathbb{Z}$, then

$$\int x^m\,(f + g\,x)\,\left(a + c\,x^2\right)^p\,dx \longrightarrow f\int x^m\,\left(a + c\,x^2\right)^p\,dx + g\int x^{m+1}\,\left(a + c\,x^2\right)^p\,dx$$

Program code:

```
Int[x_^m_.*(f_+g_.*x_)*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  f*Int[x^m*(a+c*x^2)^p,x] + g*Int[x^(m+1)*(a+c*x^2)^p,x] /;
FreeQ[{a,c,f,g,p},x] && IntegerQ[m] && Not[IntegerQ[2*p]]
```

2: $\int (e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $p\in\mathbb{Z}\,\land\,(p>0\,\lor\,a=0\land m\in\mathbb{Z})$

Derivation: Algebraic expansion

Rule 1.2.1.3.2: If $p\in\mathbb{Z}\,\land\,(p>0\,\lor\,a=0\land m\in\mathbb{Z})$, then

$$\int (e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow\,\int \text{ExpandIntegrand}\left[(e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p,\,x\right]dx$$

Program code:

```
Int[(e_.*x_)^m_.*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(e*x)^m*(f+g*x)*(a+b*x+c*x^2)^p,x],x] /;
FreeQ[{a,b,c,e,f,g,m},x] && IntegerQ[p] && (GtQ[p,0] || EqQ[a,0] && IntegerQ[m])
```

```
Int[(e_.*x_)^m_.*(f_.+g_.*x_)*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(e*x)^m*(f+g*x)*(a+c*x^2)^p,x],x] /;
FreeQ[{a,c,e,f,g,m},x] && IGtQ[p,0]
```

3: $\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c==0\,\land\,m+2\,p+3==0\,\land\,2\,c\,f-b\,g==0$

- Derivation: Quadratic recurrence 2a with $2\,c\,f-b\,g==0$: square quadratic recurrence 3b with $m+2\,p+3==0$

- Rule 1.2.1.3.3: If $b^2-4\,a\,c==0\,\land\,m+2\,p+3==0\,\land\,2\,c\,f-b\,g==0$, then

$$\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow\,-\frac{f\,g\,(d+e\,x)^{m+1}\,\left(a+b\,x+c\,x^2\right)^{p+1}}{b\,(p+1)\,(e\,f-d\,g)}$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*(f_+g_.*x_)*(a_+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  -f*g*(d+e*x)^(m+1)*(a+b*x+c*x^2)^(p+1)/(b*(p+1)*(e*f-d*g)) /;
FreeQ[{a,b,c,d,e,f,g,m,p},x] && EqQ[b^2-4*a*c,0] && EqQ[m+2*p+3,0] && EqQ[2*c*f-b*g,0]
```

4: $\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $2\,c\,f-b\,g==0\,\land\,p<-1\,\land\,m>0$

- Derivation: Integration by parts

- Basis: If $2\,c\,f-b\,g==0$, then $\partial_x\frac{g\,\left(a+b\,x+c\,x^2\right)^{p+1}}{2\,c\,(p+1)}==(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p$

- Rule 1.2.1.3.4: If $2\,c\,f-b\,g==0\,\land\,p<-1\,\land\,m>0$, then

$$\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow\,\frac{g\,(d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^{p+1}}{2\,c\,(p+1)}-\frac{e\,g\,m}{2\,c\,(p+1)}\int (d+e\,x)^{m-1}\,\left(a+b\,x+c\,x^2\right)^{p+1}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  g*(d+e*x)^m*(a+b*x+c*x^2)^(p+1)/(2*c*(p+1)) -
  e*g*m/(2*c*(p+1))*Int[(d+e*x)^(m-1)*(a+b*x+c*x^2)^(p+1),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[2*c*f-b*g,0] && LtQ[p,-1] && GtQ[m,0]
```

5. $\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0$

1: $\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c==0\,\land\,m+2\,p+3==0\,\land\,2\,c\,f-b\,g\neq 0\,\land\,2\,c\,d-b\,e\neq 0$

Derivation: Algebraic expansion

Basis: $f+g\,x==\frac{(2\,c\,f-b\,g)\,(d+e\,x)}{2\,c\,d-b\,e}-\frac{(e\,f-d\,g)\,(b+2\,c\,x)}{2\,c\,d-b\,e}$

Rule 1.2.1.3.5: If $b^2-4\,a\,c==0\,\land\,m+2\,p+3==0\,\land\,2\,c\,f-b\,g\neq 0\,\land\,2\,c\,d-b\,e\neq 0$, then

$$\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow$$
$$-\frac{2\,c\,(e\,f-d\,g)\,(d+e\,x)^{m+1}\,\left(a+b\,x+c\,x^2\right)^{p+1}}{(p+1)\,(2\,c\,d-b\,e)^2}+\frac{2\,c\,f-b\,g}{2\,c\,d-b\,e}\int (d+e\,x)^{m+1}\,\left(a+b\,x+c\,x^2\right)^p\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*(f_.+g_.*x_)*(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  -2*c*(e*f-d*g)*(d+e*x)^(m+1)*(a+b*x+c*x^2)^(p+1)/((p+1)*(2*c*d-b*e)^2) +
  (2*c*f-b*g)/(2*c*d-b*e)*Int[(d+e*x)^(m+1)*(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,f,g,m,p},x] && EqQ[b^2-4*a*c,0] && EqQ[m+2*p+3,0] && NeQ[2*c*f-b*g,0] && NeQ[2*c*d-b*e,0]
```

2: $\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c==0$

Derivation: Piecewise constant extraction

Basis: If $b^2-4\,a\,c==0$, then $\partial_x \frac{\left(a+b\,x+c\,x^2\right)^p}{\left(\frac{b}{2}+c\,x\right)^{2\,p}}==0$

Rule 1.2.1.3.6: If $b^2-4\,a\,c==0$, then

$$\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx \to \frac{\left(a+b\,x+c\,x^2\right)^{\text{FracPart}[p]}}{c^{\text{IntPart}[p]}\,\left(\frac{b}{2}+c\,x\right)^{2\,\text{FracPart}[p]}}\int (d+e\,x)^m\,(f+g\,x)\,\left(\frac{b}{2}+c\,x\right)^{2\,p}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (a+b*x+c*x^2)^FracPart[p]/(c^IntPart[p]*(b/2+c*x)^(2*FracPart[p]))*Int[(d+e*x)^m*(f+g*x)*(b/2+c*x)^(2*p),x] /;
FreeQ[{a,b,c,d,e,f,g,m},x] && EqQ[b^2-4*a*c,0]
```

6: $\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,p\in\mathbb{Z}\,\land\,(p>0\,\lor\,a==0\land m\in\mathbb{Z})$

Derivation: Algebraic expansion

Rule 1.2.1.3.6: If $b^2-4\,a\,c\neq 0\,\land\,p\in\mathbb{Z}^+$, then

$$\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx \to \int \text{ExpandIntegrand}\left[(d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p,\,x\right]dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*(f+g*x)*(a+b*x+c*x^2)^p,x],x] /;
FreeQ[{a,b,c,d,e,f,g,m},x] && NeQ[b^2-4*a*c,0] && IntegerQ[p] && (GtQ[p,0] || EqQ[a,0] && IntegerQ[m])
```

```
Int[(d_.+e_.*x_)^m_.*(f_.+g_.*x_)*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*(f+g*x)*(a+c*x^2)^p,x],x] /;
FreeQ[{a,c,d,e,f,g,m},x] && IGtQ[p,0]
```

7. $\int (d+e\,x)\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0$

1: $\int \frac{(d+e\,x)\,(f+g\,x)}{a+b\,x+c\,x^2}\,dx$ when $b^2-4\,a\,c\neq 0$

Derivation: Algebraic expansion

–

Rule 1.2.1.3.7.1: If $b^2-4\,a\,c\neq 0$, then

$$\int \frac{(d+e\,x)\,(f+g\,x)}{a+b\,x+c\,x^2}\,dx \longrightarrow \frac{e\,g\,x}{c}+\frac{1}{c}\int \frac{c\,d\,f-a\,e\,g+(c\,e\,f+c\,d\,g-b\,e\,g)\,x}{a+b\,x+c\,x^2}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)*(f_+g_.*x_)/(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  e*g*x/c + 1/c*Int[(c*d*f-a*e*g+(c*e*f+c*d*g-b*e*g)*x)/(a+b*x+c*x^2),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b^2-4*a*c,0]
```

```
Int[(d_.+e_.*x_)*(f_+g_.*x_)/(a_+c_.*x_^2),x_Symbol] :=
  e*g*x/c + 1/c*Int[(c*d*f-a*e*g+c*(e*f+d*g)*x)/(a+c*x^2),x] /;
FreeQ[{a,c,d,e,f,g},x]
```

2: $\int (d+e x) (f+g x) \left(a+b x+c x^2\right)^p \, dx$ when $b^2-4 a c \neq 0 \land b^2 e g (p+2)-2 a c e g+c (2 c d f-b (e f+d g)) (2 p+3)=0 \land p \neq -1$

Derivation: ???

Note: If $b^2-4 a c \neq 0 \land b^2 e g (p+2)-2 a c e g+c (2 c d f-b (e f+d g)) (2 p+3)=0$, then $p \neq -\frac{3}{2}$.

Rule 1.2.1.3.7.2: If $b^2-4 a c \neq 0 \land b^2 e g (p+2)-2 a c e g+c (2 c d f-b (e f+d g)) (2 p+3)=0 \land p \neq -1$, then

$$\int (d+e x) (f+g x) \left(a+b x+c x^2\right)^p \, dx \longrightarrow -\frac{(b e g (p+2)-c (e f+d g) (2 p+3)-2 c e g (p+1) x) \left(a+b x+c x^2\right)^{p+1}}{2 c^2 (p+1) (2 p+3)}$$

Program code:

```
Int[(d_.+e_.*x_)*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  -(b*e*g*(p+2)-c*(e*f+d*g)*(2*p+3)-2*c*e*g*(p+1)*x)*(a+b*x+c*x^2)^(p+1)/(2*c^2*(p+1)*(2*p+3)) /;
FreeQ[{a,b,c,d,e,f,g,p},x] && NeQ[b^2-4*a*c,0] && EqQ[b^2*e*g*(p+2)-2*a*c*e*g+c*(2*c*d*f-b*(e*f+d*g))*(2*p+3),0] && NeQ[p,-1]
```

```
Int[(d_.+e_.*x_)*(f_.+g_.*x_)*(a_+c_.*x_^2)^p_,x_Symbol] :=
  ((e*f+d*g)*(2*p+3)+2*e*g*(p+1)*x)*(a+c*x^2)^(p+1)/(2*c*(p+1)*(2*p+3)) /;
FreeQ[{a,c,d,e,f,g,p},x] && EqQ[a*e*g-c*d*f*(2*p+3),0] && NeQ[p,-1]
```

3: $\int (d+e x) (f+g x) \left(a+b x+c x^2\right)^p \, dx$ when $b^2-4 a c \neq 0 \land p<-1$

Derivation: ???

Rule 1.2.1.3.7.3: If $b^2-4 a c \neq 0 \land p<-1$, then

$$\int (d+e x) (f+g x) \left(a+b x+c x^2\right)^p \, dx \longrightarrow$$

$$-\left(\left(\left(2 a c (e f+d g)-b (c d f+a e g)-\left(b^2 e g-b c (e f+d g)+2 c (c d f-a e g)\right) x\right) \left(a+b x+c x^2\right)^{p+1}\right) \Big/ \left(c (p+1) \left(b^2-4 a c\right)\right)\right)-\frac{b^2 e g (p+2)-2 a c e g+c \left(2 c d f-b (e f+d g)\right) (2 p+3)}{c (p+1) \left(b^2-4 a c\right)} \int \left(a+b x+c x^2\right)^{p+1} \, dx$$

Program code:

```
Int[(d_.+e_.*x_)*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  -(2*a*c*(e*f+d*g)-b*(c*d*f+a*e*g)-(b^2*e*g-b*c*(e*f+d*g)+2*c*(c*d*f-a*e*g))*x)*(a+b*x+c*x^2)^(p+1)/(c*(p+1)*(b^2-4*a*c)) -
  (b^2*e*g*(p+2)-2*a*c*e*g+c*(2*c*d*f-b*(e*f+d*g))*(2*p+3))/(c*(p+1)*(b^2-4*a*c))*Int[(a+b*x+c*x^2)^(p+1),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b^2-4*a*c,0] && LtQ[p,-1]
```

```
Int[(d_.+e_.*x_)*(f_.+g_.*x_)*(a_+c_.*x_^2)^p_,x_Symbol] :=
  (a*(e*f+d*g)-(c*d*f-a*e*g)*x)*(a+c*x^2)^(p+1)/(2*a*c*(p+1)) -
  (a*e*g-c*d*f*(2*p+3))/(2*a*c*(p+1))*Int[(a+c*x^2)^(p+1),x] /;
FreeQ[{a,c,d,e,f,g},x] && LtQ[p,-1]
```

4: $\int (d+e x) (f+g x) \left(a+b x+c x^2\right)^p \, dx$ when $b^2-4 a c \neq 0 \land p \not< -1$

Derivation: ???

Rule 1.2.1.3.7.4: If $b^2-4 a c \neq 0 \land p \not< -1$, then

$$\int (d+e x) (f+g x) \left(a+b x+c x^2\right)^p \, dx \longrightarrow$$

$$-\frac{\left(b e g (p+2)-c (e f+d g) (2 p+3)-2 c e g (p+1) x\right) \left(a+b x+c x^2\right)^{p+1}}{2 c^2 (p+1) (2 p+3)}+$$

$$\frac{b^2 e g (p+2) - 2 a c e g + c \left(2 c d f - b \left(e f + d g\right)\right) (2 p + 3)}{2 c^2 (2 p + 3)} \int \left(a + b x + c x^2\right)^p dx$$

Program code:

```
Int[(d_.+e_.*x_)*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  -(b*e*g*(p+2)-c*(e*f+d*g)*(2*p+3)-2*c*e*g*(p+1)*x)*(a+b*x+c*x^2)^(p+1)/(2*c^2*(p+1)*(2*p+3)) +
  (b^2*e*g*(p+2)-2*a*c*e*g+c*(2*c*d*f-b*(e*f+d*g))*(2*p+3))/(2*c^2*(2*p+3))*Int[(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,f,g,p},x] && NeQ[b^2-4*a*c,0] && Not[LeQ[p,-1]]
```

```
Int[(d_.+e_.*x_)*(f_.+g_.*x_)*(a_+c_.*x_^2)^p_,x_Symbol] :=
  ((e*f+d*g)*(2*p+3)+2*e*g*(p+1)*x)*(a+c*x^2)^(p+1)/(2*c*(p+1)*(2*p+3)) -
  (a*e*g-c*d*f*(2*p+3))/(c*(2*p+3))*Int[(a+c*x^2)^p,x] /;
FreeQ[{a,c,d,e,f,g,p},x] && Not[LeQ[p,-1]]
```

8. $\int (d+e x)^m (f+g x) \left(a+b x+c x^2\right)^p dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 == 0$

1. $\int (d+e x)^m (f+g x) \left(a+b x+c x^2\right)^p dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 == 0 \land p \in \mathbb{Z}$

1: $\int (e x)^m (f+g x) \left(b x+c x^2\right)^p dx$ when $p \in \mathbb{Z}$

Derivation: Algebraic simplification

Rule 1.2.1.2.8.1.1: If $p \in \mathbb{Z}$, then

$$\int (e x)^m (f+g x) \left(b x+c x^2\right)^p dx \longrightarrow \frac{1}{e^p} \int (e x)^{m+p} (f+g x) (b+c x)^p dx$$

Program code:

```
Int[(e_.*x_)^m_.*(f_.+g_.*x_)*(b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  1/e^p*Int[(e*x)^(m+p)*(f+g*x)*(b+c*x)^p,x] /;
FreeQ[{b,c,e,f,g,m},x] && IntegerQ[p]
```

2: $\int (d+e x)^m (f+g x) \left(a+b x+c x^2\right)^p dx$ when $b^2-4 a c\neq 0 \land c d^2-b d e+a e^2==0 \land p\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $c d^2-b d e+a e^2==0$, then $a+b x+c x^2==(d+e x)\left(\frac{a}{d}+\frac{c x}{e}\right)$

Rule 1.2.1.3.8.1.2: If $b^2-4 a c\neq 0 \land c d^2-b d e+a e^2==0 \land p\in\mathbb{Z}$, then

$$\int (d+e x)^m (f+g x) \left(a+b x+c x^2\right)^p dx \to \int (d+e x)^{m+p} (f+g x) \left(\frac{a}{d}+\frac{c x}{e}\right)^p dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[(d+e*x)^(m+p)*(f+g*x)*(a/d+c/e*x)^p,x] /;
FreeQ[{a,b,c,d,e,f,g,m},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[p]
```

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[(d+e*x)^(m+p)*(f+g*x)*(a/d+c/e*x)^p,x] /;
FreeQ[{a,c,d,e,f,g,m},x] && EqQ[c*d^2+a*e^2,0] && (IntegerQ[p] || GtQ[a,0] && GtQ[d,0] && EqQ[m+p,0])
```

2. $\int (d+e x)^m (f+g x) \left(a+b x+c x^2\right)^p dx$ when $b^2-4 a c\neq 0 \land c d^2-b d e+a e^2==0 \land p\notin\mathbb{Z}$

0: $\int (d+e x)^m (f+g x) \left(a+b x+c x^2\right)^p dx$ when $b^2-4 a c\neq 0 \land c d^2-b d e+a e^2==0 \land p\notin\mathbb{Z} \land m\in\mathbb{Z}^-$?? ???

Derivation: Algebraic simplification

Basis: If $c d^2-b d e+a e^2==0$, then $d+e x==\frac{d e \left(a+b x+c x^2\right)}{a e+c d x}$

Basis: If $c d^2+a e^2==0$, then $d+e x==\frac{d^2 \left(a+c x^2\right)}{a (d-e x)}$

Rule 1.2.1.3.8.2.0: If $b^2-4 a c\neq 0 \land c d^2-b d e+a e^2==0 \land m\in\mathbb{Z}^-$, then

$$\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx\;\longrightarrow\;d^m\,e^m\int\frac{(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^{m+p}}{(a\,e+c\,d\,x)^m}\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  d^m*e^m*Int[(f+g*x)*(a+b*x+c*x^2)^(m+p)/(a*e+c*d*x)^m,x] /;
FreeQ[{a,b,c,d,e,f,g,p},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[2*p]] && ILtQ[m,0]
```

```
Int[x_*(d_+e_.*x_)^m_*(a_+c_.*x_^2)^p_,x_Symbol] :=
  d^m*e^m*Int[x*(a+c*x^2)^(m+p)/(a*e+c*d*x)^m,x] /;
FreeQ[{a,c,d,e,p},x] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && ILtQ[m,0]          && EqQ[m,-1] && Not[ILtQ[p-1/2,0]]
```

1: $\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2=0\,\land\,m\,\left(g\,(c\,d-b\,e)+c\,e\,f\right)+e\,(p+1)\,\left(2\,c\,f-b\,g\right)=0$

Derivation: Quadratic recurrence 3a with $c\,d^2-b\,d\,e+a\,e^2=0$ and $m\,(g\,(c\,d-b\,e)+c\,e\,f)+e\,(p+1)\,(2\,c\,f-b\,g)=0$

Note: If $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2=0\,\land\,m\,(g\,(c\,d-b\,e)+c\,e\,f)+e\,(p+1)\,(2\,c\,f-b\,g)=0$, then $m+2\,p+2\neq 0$.

Rule 1.2.1.3.8.2.1: If $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2=0\,\land\,m\,(g\,(c\,d-b\,e)+c\,e\,f)+e\,(p+1)\,(2\,c\,f-b\,g)=0$, then

$$\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx\;\longrightarrow\;\frac{g\,(d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^{p+1}}{c\,(m+2\,p+2)}$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  g*(d+e*x)^m*(a+b*x+c*x^2)^(p+1)/(c*(m+2*p+2)) /;
FreeQ[{a,b,c,d,e,f,g,m,p},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && EqQ[m*(g*(c*d-b*e)+c*e*f)+e*(p+1)*(2*c*f-b*g),0]
```

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)*(a_+c_.*x_^2)^p_,x_Symbol] :=
  g*(d+e*x)^m*(a+c*x^2)^(p+1)/(c*(m+2*p+2)) /;
FreeQ[{a,c,d,e,f,g,m,p},x] && EqQ[c*d^2+a*e^2,0] && EqQ[m*(d*g+e*f)+2*e*f*(p+1),0]
```

2: $\int (d+e x)^m (f+g x) \left(a+b x+c x^2\right)^p dx$ when $b^2-4 a c\neq 0 \wedge c d^2-b d e+a e^2=0 \wedge p<-1 \wedge m>0$

Derivation: Quadratic recurrence 3a with $c d^2-b d e+a e^2=0$: special quadratic recurrence 2b

Note: If $b^2-4 a c\neq 0 \wedge c d^2-b d e+a e^2=0$, then $2 c d-b e\neq 0$.

Rule 1.2.1.3.8.2.2: If $b^2-4 a c\neq 0 \wedge c d^2-b d e+a e^2=0 \wedge p<-1 \wedge m>0$, then

$$\int (d+e x)^m (f+g x) \left(a+b x+c x^2\right)^p dx \longrightarrow$$
$$\frac{(g (c d-b e)+c e f) (d+e x)^m \left(a+b x+c x^2\right)^{p+1}}{c (p+1) (2 c d-b e)}-\frac{e (m (g (c d-b e)+c e f)+e (p+1) (2 c f-b g))}{c (p+1) (2 c d-b e)}\int (d+e x)^{m-1} \left(a+b x+c x^2\right)^{p+1} dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (g*(c*d-b*e)+c*e*f)*(d+e*x)^m*(a+b*x+c*x^2)^(p+1)/(c*(p+1)*(2*c*d-b*e)) -
  e*(m*(g*(c*d-b*e)+c*e*f)+e*(p+1)*(2*c*f-b*g))/(c*(p+1)*(2*c*d-b*e))*
    Int[(d+e*x)^(m-1)*(a+b*x+c*x^2)^(p+1),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && LtQ[p,-1] && GtQ[m,0]
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_+c_.*x_^2)^p_,x_Symbol] :=
  (d*g+e*f)*(d+e*x)^m*(a+c*x^2)^(p+1)/(2*c*d*(p+1)) -
  e*(m*(d*g+e*f)+2*e*f*(p+1))/(2*c*d*(p+1))*Int[(d+e*x)^(m-1)*(a+c*x^2)^(p+1),x] /;
FreeQ[{a,c,d,e,f,g},x] && EqQ[c*d^2+a*e^2,0] && LtQ[p,-1] && GtQ[m,0]
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (g*(c*d-b*e)+c*e*f)*(d+e*x)^m*(a+b*x+c*x^2)^(p+1)/(c*(p+1)*(2*c*d-b*e)) -
  e*(m*(g*(c*d-b*e)+c*e*f)+e*(p+1)*(2*c*f-b*g))/(c*(p+1)*(2*c*d-b*e))*
    Int[(d+e*x)^Simplify[m-1]*(a+b*x+c*x^2)^Simplify[p+1],x] /;
FreeQ[{a,b,c,d,e,f,g,m,p},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && SumSimplerQ[p,1] && SumSimplerQ[m,-1] && NeQ[p,-1]
```

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)*(a_+c_.*x_^2)^p_,x_Symbol] :=
  (d*g+e*f)*(d+e*x)^m*(a+c*x^2)^(p+1)/(2*c*d*(p+1)) -
  e*(m*(d*g+e*f)+2*e*f*(p+1))/(2*c*d*(p+1))*Int[(d+e*x)^Simplify[m-1]*(a+c*x^2)^Simplify[p+1],x] /;
FreeQ[{a,c,d,e,f,g,m,p},x] && EqQ[c*d^2+a*e^2,0] && SumSimplerQ[p,1] && SumSimplerQ[m,-1] && NeQ[p,-1] && Not[IGtQ[m,0]]
```

3: $\int (d+e x)^m (f+g x) \left(a+b x+c x^2\right)^p dx$ when $b^2-4 a c\neq 0 \land c d^2-b d e+a e^2=0 \land (m\leq -1 \lor m+2 p+2=0) \land m+p+1\neq 0$

Derivation: Quadratic recurrence 3a with $c d^2-b d e+a e^2=0$

Note: If $b^2-4 a c\neq 0 \land c d^2-b d e+a e^2=0$, then $2 c d-b e\neq 0$.

Rule 1.2.1.3.8.2.3: If $b^2-4 a c\neq 0 \land c d^2-b d e+a e^2=0 \land (m\leq -1 \lor m+2 p+2=0) \land m+p+1\neq 0$, then

$$\int (d+e x)^m (f+g x) \left(a+b x+c x^2\right)^p dx \longrightarrow$$
$$\frac{(d g-e f) (d+e x)^m \left(a+b x+c x^2\right)^{p+1}}{(2 c d-b e) (m+p+1)}+\frac{m (g (c d-b e)+c e f)+e (p+1) (2 c f-b g)}{e (2 c d-b e) (m+p+1)}\int (d+e x)^{m+1} \left(a+b x+c x^2\right)^p dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (d*g-e*f)*(d+e*x)^m*(a+b*x+c*x^2)^(p+1)/((2*c*d-b*e)*(m+p+1)) +
  (m*(g*(c*d-b*e)+c*e*f)+e*(p+1)*(2*c*f-b*g))/(e*(2*c*d-b*e)*(m+p+1))*Int[(d+e*x)^(m+1)*(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,f,g,m,p},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] &&
  (LtQ[m,-1] && Not[IGtQ[m+p+1,0]] || LtQ[m,0] && LtQ[p,-1] || EqQ[m+2*p+2,0]) && NeQ[m+p+1,0]
```

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)*(a_+c_.*x_^2)^p_,x_Symbol] :=
  (d*g-e*f)*(d+e*x)^m*(a+c*x^2)^(p+1)/(2*c*d*(m+p+1)) +
  (m*(g*c*d+c*e*f)+2*e*c*f*(p+1))/(e*(2*c*d)*(m+p+1))*Int[(d+e*x)^(m+1)*(a+c*x^2)^p,x] /;
FreeQ[{a,c,d,e,f,g,m,p},x] && EqQ[c*d^2+a*e^2,0] &&
  (LtQ[m,-1] && Not[IGtQ[m+p+1,0]] || LtQ[m,0] && LtQ[p,-1] || EqQ[m+2*p+2,0]) && NeQ[m+p+1,0]
```

4: $\int (d+e\,x)^m\,(f+g\,x)\,(a+b\,x+c\,x^2)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,m+2\,p+2\neq 0$

Derivation: Quadratic recurrence 3a with $c\,d^2-b\,d\,e+a\,e^2==0$

Rule 1.2.1.3.8.2.4: If $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,m+2\,p+2\neq 0$, then

$$\int (d+e\,x)^m\,(f+g\,x)\,(a+b\,x+c\,x^2)^p\,dx\,\longrightarrow$$
$$\frac{g\,(d+e\,x)^m\,(a+b\,x+c\,x^2)^{p+1}}{c\,(m+2\,p+2)}+\frac{m\,(g\,(c\,d-b\,e)+c\,e\,f)+e\,(p+1)\,(2\,c\,f-b\,g)}{c\,e\,(m+2\,p+2)}\int (d+e\,x)^m\,(a+b\,x+c\,x^2)^p\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  g*(d+e*x)^m*(a+b*x+c*x^2)^(p+1)/(c*(m+2*p+2)) +
  (m*(g*(c*d-b*e)+c*e*f)+e*(p+1)*(2*c*f-b*g))/(c*e*(m+2*p+2))*Int[(d+e*x)^m*(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,f,g,m,p},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && NeQ[m+2*p+2,0] && (NeQ[m,2] || EqQ[d,0])
```

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)*(a_+c_.*x_^2)^p_,x_Symbol] :=
  g*(d+e*x)^m*(a+c*x^2)^(p+1)/(c*(m+2*p+2)) +
  (m*(d*g+e*f)+2*e*f*(p+1))/(e*(m+2*p+2))*Int[(d+e*x)^m*(a+c*x^2)^p,x] /;
FreeQ[{a,c,d,e,f,g,m,p},x] && EqQ[c*d^2+a*e^2,0] && NeQ[m+2*p+2,0] && NeQ[m,2]
```

5. $\int x^2 (f+g x) (a+c x^2)^p \, dx$ when $a g^2+f^2 c == 0$

1: $\int x^2 (f+g x) (a+c x^2)^p \, dx$ when $a g^2+f^2 c == 0 \land p<-2$

Derivation: Quadratic recurrence 2a

Rule 1.2.1.3.8.2.5.1: If $a g^2+f^2 c == 0 \land p<-2$, then

$$\int x^2 (f+g x) (a+c x^2)^p \, dx \longrightarrow \frac{x^2 (a g-c f x) (a+c x^2)^{p+1}}{2 a c (p+1)}-\frac{1}{2 a c (p+1)} \int x (2 a g-c f (2 p+5) x) (a+c x^2)^{p+1} \, dx$$

Program code:

```
Int[x_^2*(f_+g_.*x_)*(a_+c_.*x_^2)^p_,x_Symbol] :=
  x^2*(a*g-c*f*x)*(a+c*x^2)^(p+1)/(2*a*c*(p+1)) -
  1/(2*a*c*(p+1))*Int[x*Simp[2*a*g-c*f*(2*p+5)*x,x]*(a+c*x^2)^(p+1),x] /;
FreeQ[{a,c,f,g},x] && EqQ[a*g^2+f^2*c,0] && LtQ[p,-2]
```

2: $\int x^2 (f+g x) (a+c x^2)^p \, dx$ when $a g^2 + f^2 c == 0$

Derivation: Algebraic expansion

Basis: $x^2 (f+g x) == \frac{(f+g x) (a+c x^2)}{c} - \frac{a (f+g x)}{c}$

Rule 1.2.1.3.8.2.5.2: If $a g^2 + f^2 c == 0$, then

$$\int x^2 (f+g x) (a+c x^2)^p \, dx \longrightarrow \frac{1}{c} \int (f+g x) (a+c x^2)^{p+1} \, dx - \frac{a}{c} \int (f+g x) (a+c x^2)^p \, dx$$

Program code:

```
Int[x_^2*(f_+g_.*x_)*(a_+c_.*x_^2)^p_,x_Symbol] :=
  1/c*Int[(f+g*x)*(a+c*x^2)^(p+1),x] - a/c*Int[(f+g*x)*(a+c*x^2)^p,x] /;
FreeQ[{a,c,f,g,p},x] && EqQ[a*g^2+f^2*c,0]
```

?: $\int (d+e x)^m (f+g x) (a+b x+c x^2)^p \, dx$ when $b^2 - 4 a c \neq 0 \land c f^2 - b f g + a g^2 == 0 \land p \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $c f^2 - b f g + a g^2 == 0$, then $a + b x + c x^2 == (f+g x) \left(\frac{a}{f} + \frac{c x}{g}\right)$

Rule 1.2.1.3.8.1.2: If $b^2 - 4 a c \neq 0 \land c f^2 - b f g + a g^2 == 0 \land p \in \mathbb{Z}$, then

$$\int (d+e x)^m (f+g x) (a+b x+c x^2)^p \, dx \longrightarrow \int (d+e x)^m (f+g x)^{p+1} \left(\frac{a}{f} + \frac{c x}{g}\right)^p dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[(d+e*x)^m*(f+g*x)^(p+1)*(a/f+c/g*x)^p,x] /;
FreeQ[{a,b,c,d,e,f,g,m},x] && NeQ[b^2-4*a*c,0] && EqQ[c*f^2-b*f*g+a*g^2,0] && IntegerQ[p]
```

```
Int[(d_+e_.*x_)^m_*(f_.+g_.*x_)*(a_.+c_.*x_^2)^p_.,x_Symbol] :=
  Int[(d+e*x)^m*(f+g*x)^(p+1)*(a/f+c/g*x)^p,x] /;
FreeQ[{a,c,d,e,f,g,m},x] && EqQ[c*f^2+a*g^2,0] && (IntegerQ[p] || GtQ[a,0] && GtQ[f,0] && EqQ[p,-1])
```

9: $\int \frac{(d+e x)^m (f+g x)}{a+b x+c x^2} dx$ when $b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge m \in \mathbb{Z}$

Derivation: Algebraic expansion

Rule 1.2.1.3.9: If $b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge m \in \mathbb{Z}$, then

$$\int \frac{(d+e x)^m (f+g x)}{a+b x+c x^2} dx \longrightarrow \int \text{ExpandIntegrand}\left[\frac{(d+e x)^m (f+g x)}{a+b x+c x^2}, x\right] dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)/(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*(f+g*x)/(a+b*x+c*x^2),x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[m]
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)/(a_.+c_.*x_^2),x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m*(f+g*x)/(a+c*x^2),x],x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[c*d^2+a*e^2,0] && IntegerQ[m]
```

10. $\int (d+e x)^m (f+g x) \left(a+b x+c x^2\right)^p dx$ when $b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge m+2 p+3 == 0$

1: $\int (d+e x)^m (f+g x) \left(a+b x+c x^2\right)^p dx$ when $b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge m+2 p+3 == 0 \wedge b (e f+d g)-2 (c d f+a e g) == 0$

Derivation: Quadratic recurrence 3b

Rule 1.2.1.3.10.1: If
$b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge m+2 p+3 == 0 \wedge p \neq -1 \wedge b (e f+d g)-2 (c d f+a e g) == 0$, then

$$\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx \longrightarrow -\frac{(e\,f-d\,g)\,(d+e\,x)^{m+1}\,\left(a+b\,x+c\,x^2\right)^{p+1}}{2\,(p+1)\,\left(c\,d^2-b\,d\,e+a\,e^2\right)}$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  -(e*f-d*g)*(d+e*x)^(m+1)*(a+b*x+c*x^2)^(p+1)/(2*(p+1)*(c*d^2-b*d*e+a*e^2)) /;
FreeQ[{a,b,c,d,e,f,g,m,p},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && EqQ[Simplify[m+2*p+3],0] && EqQ[b*(e*f+d*g)-2*(c*d*f+a*e*g),
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  -(e*f-d*g)*(d+e*x)^(m+1)*(a+c*x^2)^(p+1)/(2*(p+1)*(c*d^2+a*e^2)) /;
FreeQ[{a,c,d,e,f,g,m,p},x] && NeQ[c*d^2+a*e^2,0] && EqQ[Simplify[m+2*p+3],0] && EqQ[c*d*f+a*e*g,0]
```

2: $\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,m+2\,p+3=0\,\land\,p<-1$

Derivation: Quadratic recurrence 2a

Rule 1.2.1.3.10.2: If $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,m+2\,p+3=0\,\land\,p<-1$, then

$$\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx \longrightarrow$$

$$\frac{(d+e\,x)^m\,\left(a+b\,x+c\,x^2\right)^{p+1}\,(b\,f-2\,a\,g+(2\,c\,f-b\,g)\,x)}{(p+1)\,\left(b^2-4\,a\,c\right)}+\frac{m\,(b\,(e\,f+d\,g)-2\,(c\,d\,f+a\,e\,g))}{(p+1)\,\left(b^2-4\,a\,c\right)}\int (d+e\,x)^{m-1}\,\left(a+b\,x+c\,x^2\right)^{p+1}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (d+e*x)^m*(a+b*x+c*x^2)^(p+1)*(b*f-2*a*g+(2*c*f-b*g)*x)/((p+1)*(b^2-4*a*c)) -
  m*(b*(e*f+d*g)-2*(c*d*f+a*e*g))/((p+1)*(b^2-4*a*c))*Int[(d+e*x)^(m-1)*(a+b*x+c*x^2)^(p+1),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && EqQ[Simplify[m+2*p+3],0] && LtQ[p,-1]
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_+c_.*x_^2)^p_,x_Symbol] :=
  (d+e*x)^m*(a+c*x^2)^(p+1)*(a*g-c*f*x)/(2*a*c*(p+1)) -
  m*(c*d*f+a*e*g)/(2*a*c*(p+1))*Int[(d+e*x)^(m-1)*(a+c*x^2)^(p+1),x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[c*d^2+a*e^2,0] && EqQ[Simplify[m+2*p+3],0] && LtQ[p,-1]
```

3: $\int (d+e x)^m (f+g x) \left(a+b x+c x^2\right)^p dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land m+2 p+3 == 0 \land p \nless -1$

Derivation: Quadratic recurrence 3b

Rule 1.2.1.3.10.3: If $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land m+2 p+3 == 0 \land p \nless -1$, then

$$\int (d+e x)^m (f+g x) \left(a+b x+c x^2\right)^p dx \to$$
$$-\frac{(e f-d g) (d+e x)^{m+1} \left(a+b x+c x^2\right)^{p+1}}{2 (p+1) \left(c d^2-b d e+a e^2\right)}-\frac{b (e f+d g)-2 (c d f+a e g)}{2 \left(c d^2-b d e+a e^2\right)} \int (d+e x)^{m+1} \left(a+b x+c x^2\right)^p dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  -(e*f-d*g)*(d+e*x)^(m+1)*(a+b*x+c*x^2)^(p+1)/(2*(p+1)*(c*d^2-b*d*e+a*e^2)) -
  (b*(e*f+d*g)-2*(c*d*f+a*e*g))/(2*(c*d^2-b*d*e+a*e^2))*Int[(d+e*x)^(m+1)*(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,f,g,m,p},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && EqQ[Simplify[m+2*p+3],0]
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  -(e*f-d*g)*(d+e*x)^(m+1)*(a+c*x^2)^(p+1)/(2*(p+1)*(c*d^2+a*e^2)) +
  (c*d*f+a*e*g)/(c*d^2+a*e^2)*Int[(d+e*x)^(m+1)*(a+c*x^2)^p,x] /;
FreeQ[{a,c,d,e,f,g,m,p},x] && NeQ[c*d^2+a*e^2,0] && EqQ[Simplify[m+2*p+3],0]
```

11: $\int (e\,x)^m\,(f+g\,x)\,(a+c\,x^2)^p\,dx$ when $m\notin\mathbb{Q}\,\land\,p\notin\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.2.1.3.11: If $m\notin\mathbb{Q}\,\land\,p\notin\mathbb{Z}^+$, then

$$\int (e\,x)^m\,(f+g\,x)\,(a+c\,x^2)^p\,dx\,\longrightarrow\,f\int (e\,x)^m\,(a+c\,x^2)^p\,dx+\frac{g}{e}\int (e\,x)^{m+1}\,(a+c\,x^2)^p\,dx$$

Program code:

```
Int[(e_.*x_)^m_*(f_+g_.*x_)*(a_+c_.*x_^2)^p_,x_Symbol] :=
  f*Int[(e*x)^m*(a+c*x^2)^p,x] + g/e*Int[(e*x)^(m+1)*(a+c*x^2)^p,x] /;
FreeQ[{a,c,e,f,g,p},x] && Not[RationalQ[m]] && Not[IGtQ[p,0]]
```

12: $\int (d+e\,x)^m\,(f+g\,x)\,(a+b\,x+c\,x^2)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,m==p\,\land\,b\,d+a\,e==0\,\land\,c\,d+b\,e==0$

Derivation: Piecewise constant extraction

Basis: If $b\,d+a\,e==0\,\land\,c\,d+b\,e==0$, then $\partial_x\frac{(d+e\,x)^p\,(a+b\,x+c\,x^2)^p}{(a\,d+c\,e\,x^3)^p}==0$

Rule 1.2.1.3.12: If $m==p\,\land\,b\,d+a\,e==0\,\land\,c\,d+b\,e==0$, then

$$\int (d+e\,x)^m\,(f+g\,x)\,(a+b\,x+c\,x^2)^p\,dx\,\longrightarrow\,\frac{(d+e\,x)^{\text{FracPart}[p]}\,(a+b\,x+c\,x^2)^{\text{FracPart}[p]}}{(a\,d+c\,e\,x^3)^{\text{FracPart}[p]}}\int (f+g\,x)\,(a\,d+c\,e\,x^3)^p\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (d+e*x)^FracPart[p]*(a+b*x+c*x^2)^FracPart[p]/(a*d+c*e*x^3)^FracPart[p]*Int[(f+g*x)*(a*d+c*e*x^3)^p,x] /;
FreeQ[{a,b,c,d,e,f,g,m,p},x] && EqQ[m,p] && EqQ[b*d+a*e,0] && EqQ[c*d+b*e,0]
```

13. $\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,p>0$

1: $\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,p>0\,\wedge\,m<-2$

Derivation: ???

Rule 1.2.1.3.13.1: If $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,p>0\,\wedge\,m<-2$, then

$$\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow$$

$$-\frac{(d+e\,x)^{m+1}\,\left(a+b\,x+c\,x^2\right)^p}{e^2\,(m+1)\,(m+2)\,\left(c\,d^2-b\,d\,e+a\,e^2\right)}\,\cdot$$
$$\left(\left(d\,g-e\,f\,(m+2)\right)\,\left(c\,d^2-b\,d\,e+a\,e^2\right)-d\,p\,(2\,c\,d-b\,e)\,\left(e\,f-d\,g\right)-e\,\left(g\,(m+1)\,\left(c\,d^2-b\,d\,e+a\,e^2\right)+p\,(2\,c\,d-b\,e)\,\left(e\,f-d\,g\right)\right)\,x\right)-$$

$$\frac{p}{e^2\,(m+1)\,(m+2)\,\left(c\,d^2-b\,d\,e+a\,e^2\right)}\int (d+e\,x)^{m+2}\,\left(a+b\,x+c\,x^2\right)^{p-1}\cdot$$
$$\left(2\,a\,c\,e\,\left(e\,f-d\,g\right)\,(m+2)+b^2\,e\,\left(d\,g\,(p+1)-e\,f\,(m+p+2)\right)+b\,\left(a\,e^2\,g\,(m+1)-c\,d\,\left(d\,g\,(2\,p+1)-e\,f\,(m+2\,p+2)\right)\right)-\right.$$
$$\left.c\,\left(2\,c\,d\,\left(d\,g\,(2\,p+1)-e\,f\,(m+2\,p+2)\right)-e\,\left(2\,a\,e\,g\,(m+1)-b\,\left(d\,g\,(m-2\,p)+e\,f\,(m+2\,p+2)\right)\right)\right)\,x\right)\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  -(d+e*x)^(m+1)*(a+b*x+c*x^2)^p/(e^2*(m+1)*(m+2)*(c*d^2-b*d*e+a*e^2))*
    ((d*g-e*f*(m+2))*(c*d^2-b*d*e+a*e^2)-d*p*(2*c*d-b*e)*(e*f-d*g)-e*(g*(m+1)*(c*d^2-b*d*e+a*e^2)+p*(2*c*d-b*e)*(e*f-d*g))*x) -
  p/(e^2*(m+1)*(m+2)*(c*d^2-b*d*e+a*e^2))*Int[(d+e*x)^(m+2)*(a+b*x+c*x^2)^(p-1)*
    Simp[2*a*c*e*(e*f-d*g)*(m+2)+b^2*e*(d*g*(p+1)-e*f*(m+p+2))+b*(a*e^2*g*(m+1)-c*d*(d*g*(2*p+1)-e*f*(m+2*p+2)))-
      c*(2*c*d*(d*g*(2*p+1)-e*f*(m+2*p+2))-e*(2*a*e*g*(m+1)-b*(d*g*(m-2*p)+e*f*(m+2*p+2))))*x,x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] &&
  GtQ[p,0] && LtQ[m,-2] && LtQ[m+2*p,0] && Not[ILtQ[m+2*p+3,0]]
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  -(d+e*x)^(m+1)*(a+c*x^2)^p/(e^2*(m+1)*(m+2)*(c*d^2+a*e^2))*
    ((d*g-e*f*(m+2))*(c*d^2+a*e^2)-2*c*d^2*p*(e*f-d*g)-e*(g*(m+1)*(c*d^2+a*e^2)+2*c*d*p*(e*f-d*g))*x) -
  p/(e^2*(m+1)*(m+2)*(c*d^2+a*e^2))*Int[(d+e*x)^(m+2)*(a+c*x^2)^(p-1)*
    Simp[2*a*c*e*(e*f-d*g)*(m+2)-c*(2*c*d*(d*g*(2*p+1)-e*f*(m+2*p+2))-2*a*e^2*g*(m+1))*x,x],x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[c*d^2+a*e^2,0] &&
  GtQ[p,0] && LtQ[m,-2] && LtQ[m+2*p,0] && Not[ILtQ[m+2*p+3,0]]
```

2: $\int (d+e x)^m (f+g x) (a+b x+c x^2)^p \, dx$ when $b^2-4ac \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land p>0 \land m<-1 \land m+2p+1 \notin \mathbb{Z}^-$

Derivation: Quadratic recurrence 1a

Rule 1.2.1.3.13.2: If $b^2-4ac \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land p>0 \land m<-1 \land m+2p+1 \notin \mathbb{Z}^-$, then

$$\int (d+e x)^m (f+g x) (a+b x+c x^2)^p \, dx \to$$
$$\frac{(d+e x)^{m+1} (f e (m+2p+2)-g d (2p+1)+e g (m+1) x) (a+b x+c x^2)^p}{e^2 (m+1) (m+2p+2)}+$$
$$\frac{p}{e^2 (m+1) (m+2p+2)} \int (d+e x)^{m+1} (a+b x+c x^2)^{p-1} \cdot$$
$$(g (b d+2 a e+2 a e m+2 b d p)-f b e (m+2p+2)+(g (2 c d+b e+b e m+4 c d p)-2 c e f (m+2p+2)) x) \, dx$$

–

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  (d+e*x)^(m+1)*(e*f*(m+2*p+2)-d*g*(2*p+1)+e*g*(m+1)*x)*(a+b*x+c*x^2)^p/(e^2*(m+1)*(m+2*p+2)) +
  p/(e^2*(m+1)*(m+2*p+2))*Int[(d+e*x)^(m+1)*(a+b*x+c*x^2)^(p-1)*
    Simp[g*(b*d+2*a*e+2*a*e*m+2*b*d*p)-f*b*e*(m+2*p+2)+(g*(2*c*d+b*e+b*e*m+4*c*d*p)-2*c*e*f*(m+2*p+2))*x,x],x] /;
FreeQ[{a,b,c,d,e,f,g,m},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && RationalQ[p] && p>0 &&
  (LtQ[m,-1] || EqQ[p,1] || IntegerQ[p] && Not[RationalQ[m]]) && NeQ[m,-1] && Not[ILtQ[m+2*p+1,0]] &&
  (IntegerQ[m] || IntegerQ[p] || IntegersQ[2*m,2*p])
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  (d+e*x)^(m+1)*(e*f*(m+2*p+2)-d*g*(2*p+1)+e*g*(m+1)*x)*(a+c*x^2)^p/(e^2*(m+1)*(m+2*p+2)) +
  p/(e^2*(m+1)*(m+2*p+2))*Int[(d+e*x)^(m+1)*(a+c*x^2)^(p-1)*
    Simp[g*(2*a*e+2*a*e*m)+(g*(2*c*d+4*c*d*p)-2*c*e*f*(m+2*p+2))*x,x],x] /;
FreeQ[{a,c,d,e,f,g,m},x] && NeQ[c*d^2+a*e^2,0] && RationalQ[p] && p>0 &&
  (LtQ[m,-1] || EqQ[p,1] || IntegerQ[p] && Not[RationalQ[m]]) && NeQ[m,-1] && Not[ILtQ[m+2*p+1,0]] &&
  (IntegerQ[m] || IntegerQ[p] || IntegersQ[2*m,2*p])
```

3: $\int (d+e x)^m (f+g x) \left(a+b x+c x^2\right)^p dx$ when $b^2-4 a c\neq 0 \wedge c d^2-b d e+a e^2\neq 0 \wedge p>0 \wedge -1\leq m<0 \wedge m+2 p\notin \mathbb{Z}^-$

Derivation: Quadratic recurrence 1b

Rule 1.2.1.3.13.3: If $b^2-4 a c\neq 0 \wedge c d^2-b d e+a e^2\neq 0 \wedge p>0 \wedge -1\leq m<0 \wedge m+2 p\notin \mathbb{Z}^-$, then

$$\int (d+e x)^m (f+g x) \left(a+b x+c x^2\right)^p dx \longrightarrow$$

$$\left(\left((d+e x)^{m+1} (c e f (m+2 p+2)-g (c d+2 c d p-b e p)+g c e (m+2 p+1) x) \left(a+b x+c x^2\right)^p\right)/\left(c e^2 (m+2 p+1) (m+2 p+2)\right)\right)-$$
$$\frac{p}{c e^2 (m+2 p+1) (m+2 p+2)} \int (d+e x)^m \left(a+b x+c x^2\right)^{p-1}\cdot$$
$$(c e f (b d-2 a e) (m+2 p+2)+g (a e (b e-2 c d m+b e m)+b d (b e p-c d-2 c d p))+$$
$$\left(c e f (2 c d-b e) (m+2 p+2)+g \left(b^2 e^2 (p+m+1)-2 c^2 d^2 (1+2 p)-c e (b d (m-2 p)+2 a e (m+2 p+1))\right)\right) x) dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  (d+e*x)^(m+1)*(c*e*f*(m+2*p+2)-g*(c*d+2*c*d*p-b*e*p)+g*c*e*(m+2*p+1)*x)*(a+b*x+c*x^2)^p/
    (c*e^2*(m+2*p+1)*(m+2*p+2)) -
  p/(c*e^2*(m+2*p+1)*(m+2*p+2))*Int[(d+e*x)^m*(a+b*x+c*x^2)^(p-1)*
    Simp[c*e*f*(b*d-2*a*e)*(m+2*p+2)+g*(a*e*(b*e-2*c*d*m+b*e*m)+b*d*(b*e*p-c*d-2*c*d*p))+
      (c*e*f*(2*c*d-b*e)*(m+2*p+2)+g*(b^2*e^2*(p+m+1)-2*c^2*d^2*(1+2*p)-c*e*(b*d*(m-2*p)+2*a*e*(m+2*p+1))))*x,x],x] /;
FreeQ[{a,b,c,d,e,f,g,m},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] &&
  GtQ[p,0] && (IntegerQ[p] || Not[RationalQ[m]] || GeQ[m,-1] && LtQ[m,0]) && Not[ILtQ[m+2*p,0]] &&
  (IntegerQ[m] || IntegerQ[p] || IntegersQ[2*m,2*p])
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  (d+e*x)^(m+1)*(c*e*f*(m+2*p+2)-g*c*d*(2*p+1)+g*c*e*(m+2*p+1)*x)*(a+c*x^2)^p/
    (c*e^2*(m+2*p+1)*(m+2*p+2)) +
  2*p/(c*e^2*(m+2*p+1)*(m+2*p+2))*Int[(d+e*x)^m*(a+c*x^2)^(p-1)*
    Simp[f*a*c*e^2*(m+2*p+2)+a*c*d*e*g*m-(c^2*f*d*e*(m+2*p+2)-g*(c^2*d^2*(2*p+1)+a*c*e^2*(m+2*p+1)))*x,x],x] /;
FreeQ[{a,c,d,e,f,g,m},x] && NeQ[c*d^2+a*e^2,0] &&
  GtQ[p,0] && (IntegerQ[p] || Not[RationalQ[m]] || GeQ[m,-1] && LtQ[m,0]) && Not[ILtQ[m+2*p,0]] &&
  (IntegerQ[m] || IntegerQ[p] || IntegersQ[2*m,2*p])
```

14. $\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,p<-1$

1. $\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,p<-1\,\land\,m>1$

1: $\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,p<-1\,\land\,m\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.2.1.3.14.1.1: If $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,p<-1\,\land\,m\in\mathbb{Z}^+$, then

$$\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow$$

$$\int \left(a+b\,x+c\,x^2\right)^p\,\text{ExpandIntegrand}\left[(d+e\,x)^m\,(f+g\,x),\,x\right]\,dx$$

Program code:

```
Int[(d_+e_.*x_)^m_*(f_+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  Int[(a+b*x+c*x^2)^p*ExpandIntegrand[(d+e*x)^m*(f+g*x),x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && ILtQ[p,-1] && IGtQ[m,0] && RationalQ[a,b,c,d,e,f,g]
```

```
Int[(d_+e_.*x_)^m_*(f_+g_.*x_)*(a_+c_.*x_^2)^p_,x_Symbol] :=
  Int[(a+c*x^2)^p*ExpandIntegrand[(d+e*x)^m*(f+g*x),x],x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[c*d^2+a*e^2,0] && ILtQ[p,-1] && IGtQ[m,0] && RationalQ[a,c,d,e,f,g]
```

2: $\int (d+e x)^m (f+g x) \left(a+b x+c x^2\right)^p dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land p<-1 \land m>1$

Derivation: ???

Note: Although powerful, this rule results in more complicated coefficients unless $b == 0 \land d == 0$ or the parameters are all numeric.

Rule 1.2.1.3.14.1.2: If $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land p<-1 \land m>1$, then

$$\int (d+e x)^m (f+g x) \left(a+b x+c x^2\right)^p dx \longrightarrow$$

$$-\left(\left((d+e x)^{m-1} \left(a+b x+c x^2\right)^{p+1} \left(2 a c (e f+d g)-b (c d f+a e g)-\left(2 c^2 d f+b^2 e g-c (b e f+b d g+2 a e g)\right) x\right)\right) \Big/ \left(c (p+1) \left(b^2-4 a c\right)\right)\right)-\frac{1}{c (p+1) \left(b^2-4 a c\right)} \int (d+e x)^{m-2} \left(a+b x+c x^2\right)^{p+1} \cdot$$
$$\left(2 c^2 d^2 f (2 p+3)+b e g (a e (m-1)+b d (p+2))-c \left(2 a e \left(e f (m-1)+d g m\right)+b d \left(d g (2 p+3)-e f (m-2 p-4)\right)\right)+\right.$$
$$\left.e \left(b^2 e g (m+p+1)+2 c^2 d f (m+2 p+2)-c \left(2 a e g m+b (e f+d g) (m+2 p+2)\right)\right) x\right) dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  -(d+e*x)^(m-1)*(a+b*x+c*x^2)^(p+1)*(2*a*c*(e*f+d*g)-b*(c*d*f+a*e*g)-(2*c^2*d*f+b^2*e*g-c*(b*e*f+b*d*g+2*a*e*g))*x)/
    (c*(p+1)*(b^2-4*a*c)) -
  1/(c*(p+1)*(b^2-4*a*c))*Int[(d+e*x)^(m-2)*(a+b*x+c*x^2)^(p+1)*
    Simp[2*c^2*d^2*f*(2*p+3)+b*e*g*(a*e*(m-1)+b*d*(p+2))-c*(2*a*e*(e*f*(m-1)+d*g*m)+b*d*(d*g*(2*p+3)-e*f*(m-2*p-4))) +
      e*(b^2*e*g*(m+p+1)+2*c^2*d*f*(m+2*p+2)-c*(2*a*e*g*m+b*(e*f+d*g)*(m+2*p+2)))*x,x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && LtQ[p,-1] && GtQ[m,1] &&
  (EqQ[m,2] && EqQ[p,-3] && RationalQ[a,b,c,d,e,f,g] || Not[ILtQ[m+2*p+3,0]])
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  (d+e*x)^(m-1)*(a+c*x^2)^(p+1)*(a*(e*f+d*g)-(c*d*f-a*e*g)*x)/(2*a*c*(p+1)) -
  1/(2*a*c*(p+1))*Int[(d+e*x)^(m-2)*(a+c*x^2)^(p+1)*
    Simp[a*e*(e*f*(m-1)+d*g*m)-c*d^2*f*(2*p+3)+e*(a*e*g*m-c*d*f*(m+2*p+2))*x,x],x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[c*d^2+a*e^2,0] && LtQ[p,-1] && GtQ[m,1] &&
  (EqQ[d,0] || EqQ[m,2] && EqQ[p,-3] && RationalQ[a,c,d,e,f,g] || Not[ILtQ[m+2*p+3,0]])
```

2: $\int (d+e x)^m (f+g x) (a+b x+c x^2)^p \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land p<-1 \land m>0$

Derivation: Quadratic recurrence 2a

Rule 1.2.1.3.14.2: If $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land p<-1 \land m>0$, then

$$\int (d+e x)^m (f+g x) (a+b x+c x^2)^p \, dx \to$$
$$\frac{(d+e x)^m (a+b x+c x^2)^{p+1} (f b-2 a g+(2 c f-b g) x)}{(p+1) (b^2-4 a c)}+$$
$$\frac{1}{(p+1) (b^2-4 a c)} \int (d+e x)^{m-1} (a+b x+c x^2)^{p+1} \cdot$$
$$(g (2 a e m+b d (2 p+3))-f (b e m+2 c d (2 p+3))-e (2 c f-b g) (m+2 p+3) x) \, dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (d+e*x)^m*(a+b*x+c*x^2)^(p+1)*(f*b-2*a*g+(2*c*f-b*g)*x)/((p+1)*(b^2-4*a*c)) +
  1/((p+1)*(b^2-4*a*c))*Int[(d+e*x)^(m-1)*(a+b*x+c*x^2)^(p+1)*
    Simp[g*(2*a*e*m+b*d*(2*p+3))-f*(b*e*m+2*c*d*(2*p+3))-e*(2*c*f-b*g)*(m+2*p+3)*x,x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && LtQ[p,-1] && GtQ[m,0] &&
  (IntegerQ[m] || IntegerQ[p] || IntegersQ[2*m,2*p])
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_+c_.*x_^2)^p_,x_Symbol] :=
  (d+e*x)^m*(a+c*x^2)^(p+1)*(a*g-c*f*x)/(2*a*c*(p+1)) -
  1/(2*a*c*(p+1))*Int[(d+e*x)^(m-1)*(a+c*x^2)^(p+1)*Simp[a*e*g*m-c*d*f*(2*p+3)-c*e*f*(m+2*p+3)*x,x],x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[c*d^2+a*e^2,0] && LtQ[p,-1] && GtQ[m,0] &&
  (IntegerQ[m] || IntegerQ[p] || IntegersQ[2*m,2*p])
```

3: $\int (d+e x)^m (f+g x) \left(a+b x+c x^2\right)^p \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land p<-1$

Derivation: Quadratic recurrence 2b

Rule 1.2.1.3.14.3: If $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land p<-1$, then

$$\int (d+e x)^m (f+g x) \left(a+b x+c x^2\right)^p \, dx \to$$

$$\left(\left((d+e x)^{m+1} \left(f \left(b c d-b^2 e+2 a c e\right)-a g (2 c d-b e)+c \left(f (2 c d-b e)-g (b d-2 a e)\right) x\right) \left(a+b x+c x^2\right)^{p+1}\right) \Big/ \left((p+1) \left(b^2-4 a c\right) \left(c d^2-b d e+a e^2\right)\right)\right)+$$
$$\frac{1}{(p+1) \left(b^2-4 a c\right) \left(c d^2-b d e+a e^2\right)} \int (d+e x)^m \left(a+b x+c x^2\right)^{p+1} \cdot$$
$$\left(f \left(b c d e (2 p-m+2)+b^2 e^2 (p+m+2)-2 c^2 d^2 (2 p+3)-2 a c e^2 (m+2 p+3)\right)-g (a e (b e-2 c d m+b e m)-b d (3 c d-b e+2 c d p-b e p))+\right.$$
$$\left.c e \left(g (b d-2 a e)-f (2 c d-b e)\right) (m+2 p+4) x\right) dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (d+e*x)^(m+1)*(f*(b*c*d-b^2*e+2*a*c*e)-a*g*(2*c*d-b*e)+c*(f*(2*c*d-b*e)-g*(b*d-2*a*e))*x)*(a+b*x+c*x^2)^(p+1)/
    ((p+1)*(b^2-4*a*c)*(c*d^2-b*d*e+a*e^2)) +
  1/((p+1)*(b^2-4*a*c)*(c*d^2-b*d*e+a*e^2))*Int[(d+e*x)^m*(a+b*x+c*x^2)^(p+1)*
    Simp[f*(b*c*d*e*(2*p-m+2)+b^2*e^2*(p+m+2)-2*c^2*d^2*(2*p+3)-2*a*c*e^2*(m+2*p+3))-
      g*(a*e*(b*e-2*c*d*m+b*e*m)-b*d*(3*c*d-b*e+2*c*d*p-b*e*p))+
      c*e*(g*(b*d-2*a*e)-f*(2*c*d-b*e))*(m+2*p+4)*x,x],x] /;
FreeQ[{a,b,c,d,e,f,g,m},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && LtQ[p,-1] && (IntegerQ[m] || IntegerQ[p] || IntegersQ[2*m,2*p])
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_+c_.*x_^2)^p_,x_Symbol] :=
  -(d+e*x)^(m+1)*(f*a*c*e-a*g*c*d+c*(c*d*f+a*e*g)*x)*(a+c*x^2)^(p+1)/(2*a*c*(p+1)*(c*d^2+a*e^2)) +
  1/(2*a*c*(p+1)*(c*d^2+a*e^2))*Int[(d+e*x)^m*(a+c*x^2)^(p+1)*
    Simp[f*(c^2*d^2*(2*p+3)+a*c*e^2*(m+2*p+3))-a*c*d*e*g*m+c*e*(c*d*f+a*e*g)*(m+2*p+4)*x,x],x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[c*d^2+a*e^2,0] && LtQ[p,-1] && (IntegerQ[m] || IntegerQ[p] || IntegersQ[2*m,2*p])
```

15. $\int \frac{(d+e x)^m (f+g x)}{a+b x+c x^2} \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land m \notin \mathbb{Z}$

1. $\int \frac{(d+e x)^m (f+g x)}{a+b x+c x^2} \, dx$ when $b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge m \in \mathbb{Q}$

1: $\int \frac{(d+e x)^m (f+g x)}{a+b x+c x^2} \, dx$ when $b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge m \notin \mathbb{Z} \wedge m>0$

Derivation: Quadratic recurrence 3a with $p = -1$

Rule 1.2.1.3.15.1.1: If $b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge m \notin \mathbb{Z} \wedge m>0$, then

$$\int \frac{(d+e x)^m (f+g x)}{a+b x+c x^2} \, dx \longrightarrow \frac{g (d+e x)^m}{c m}+\frac{1}{c} \int \frac{(d+e x)^{m-1} (c d f-a e g+(g c d-b e g+c e f) x)}{a+b x+c x^2} \, dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)/(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  g*(d+e*x)^m/(c*m) +
  1/c*Int[(d+e*x)^(m-1)*Simp[c*d*f-a*e*g+(g*c*d-b*e*g+c*e*f)*x,x]/(a+b*x+c*x^2),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && FractionQ[m] && GtQ[m,0]
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)/(a_+c_.*x_^2),x_Symbol] :=
  g*(d+e*x)^m/(c*m) +
  1/c*Int[(d+e*x)^(m-1)*Simp[c*d*f-a*e*g+(g*c*d+c*e*f)*x,x]/(a+c*x^2),x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[c*d^2+a*e^2,0] && FractionQ[m] && GtQ[m,0]
```

2. $\int \frac{(d+e x)^m (f+g x)}{a+b x+c x^2} \, dx$ when $b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge m \notin \mathbb{Z} \wedge m<0$

1: $\int \frac{f+g x}{\sqrt{d+e x} \, (a+b x+c x^2)} \, dx$ when $b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0$

Derivation: Integration by substitution

Basis: $\frac{f+g x}{\sqrt{d+e x} \, (a+b x+c x^2)} == 2 \, \text{Subst}\left[\frac{e f-d g+g x^2}{c d^2-b d e+a e^2-(2 c d-b e) x^2+c x^4}, x, \sqrt{d+e x}\right] \partial_x \sqrt{d+e x}$

Rule 1.2.1.3.15.1.2.1: If $b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0$, then

$$\int \frac{f+g\,x}{\sqrt{d+e\,x}\,\left(a+b\,x+c\,x^2\right)}\,dx \;\longrightarrow\; 2\,\text{Subst}\left[\int \frac{e\,f-d\,g+g\,x^2}{c\,d^2-b\,d\,e+a\,e^2-(2\,c\,d-b\,e)\,x^2+c\,x^4}\,dx,\,x,\,\sqrt{d+e\,x}\right]$$

Program code:

```
Int[(f_.+g_.*x_)/(Sqrt[d_.+e_.*x_]*(a_.+b_.*x_+c_.*x_^2)),x_Symbol] :=
  2*Subst[Int[(e*f-d*g+g*x^2)/(c*d^2-b*d*e+a*e^2-(2*c*d-b*e)*x^2+c*x^4),x],x,Sqrt[d+e*x]] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0]
```

```
Int[(f_.+g_.*x_)/(Sqrt[d_.+e_.*x_]*(a_+c_.*x_^2)),x_Symbol] :=
  2*Subst[Int[(e*f-d*g+g*x^2)/(c*d^2+a*e^2-2*c*d*x^2+c*x^4),x],x,Sqrt[d+e*x]] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[c*d^2+a*e^2,0]
```

2: $\int \frac{(d+e\,x)^m\,(f+g\,x)}{a+b\,x+c\,x^2}\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,m\notin\mathbb{Z}\,\land\,m<-1$

Derivation: Quadratic recurrence 3b

Rule 1.2.1.3.15.1.2.2: If $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,m\notin\mathbb{Z}\,\land\,m<-1$, then

$$\int \frac{(d+e\,x)^m\,(f+g\,x)}{a+b\,x+c\,x^2}\,dx \;\longrightarrow\; \frac{(e\,f-d\,g)\,(d+e\,x)^{m+1}}{(m+1)\,\left(c\,d^2-b\,d\,e+a\,e^2\right)}+\frac{1}{c\,d^2-b\,d\,e+a\,e^2}\int \frac{(d+e\,x)^{m+1}\,(c\,d\,f-f\,b\,e+a\,e\,g-c\,(e\,f-d\,g)\,x)}{a+b\,x+c\,x^2}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)/(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  (e*f-d*g)*(d+e*x)^(m+1)/((m+1)*(c*d^2-b*d*e+a*e^2)) +
  1/(c*d^2-b*d*e+a*e^2)*Int[(d+e*x)^(m+1)*Simp[c*d*f-f*b*e+a*e*g-c*(e*f-d*g)*x,x]/(a+b*x+c*x^2),x] /;
FreeQ[{a,b,c,d,e,f,g,m},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && FractionQ[m] && LtQ[m,-1]
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)/(a_+c_.*x_^2),x_Symbol] :=
  (e*f-d*g)*(d+e*x)^(m+1)/((m+1)*(c*d^2+a*e^2)) +
  1/(c*d^2+a*e^2)*Int[(d+e*x)^(m+1)*Simp[c*d*f+a*e*g-c*(e*f-d*g)*x,x]/(a+c*x^2),x] /;
FreeQ[{a,c,d,e,f,g,m},x] && NeQ[c*d^2+a*e^2,0] && FractionQ[m] && LtQ[m,-1]
```

2: $\int \frac{(d+e\,x)^m\,(f+g\,x)}{a+b\,x+c\,x^2}\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,m\notin\mathbb{Q}$

Derivation: Algebraic expansion

Rule 1.2.1.3.15.2: If $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,m\notin\mathbb{Z}$, then

$$\int \frac{(d+e\,x)^m\,(f+g\,x)}{a+b\,x+c\,x^2}\,dx\;\longrightarrow\;\int (d+e\,x)^m\,\text{ExpandIntegrand}\left[\frac{f+g\,x}{a+b\,x+c\,x^2},\,x\right]dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)/(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m,(f+g*x)/(a+b*x+c*x^2),x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && Not[RationalQ[m]]
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)/(a_+c_.*x_^2),x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x)^m,(f+g*x)/(a+c*x^2),x],x] /;
FreeQ[{a,c,d,e,f,g},x] && NeQ[c*d^2+a*e^2,0] && Not[RationalQ[m]]
```

16: $\int (d+e\,x)^m\,(f+g\,x)\,(a+b\,x+c\,x^2)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,m>0\,\wedge\,m+2\,p+2\neq 0$

Derivation: Quadratic recurrence 3a

Note: The special case rule for $m=1$ and $p=-1$ eliminates the constant term $\frac{g\,d}{c}$ from the result.

Rule 1.2.1.3.16: If $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\wedge\,m>0\,\wedge\,m+2\,p+2\neq 0$, then

$$\int (d+e\,x)^m\,(f+g\,x)\,(a+b\,x+c\,x^2)^p\,dx\,\longrightarrow$$

$$\frac{g\,(d+e\,x)^m\,(a+b\,x+c\,x^2)^{p+1}}{c\,(m+2\,p+2)}+\frac{1}{c\,(m+2\,p+2)}\int (d+e\,x)^{m-1}\,(a+b\,x+c\,x^2)^p\,\cdot$$
$$(m\,(c\,d\,f-a\,e\,g)+d\,(2\,c\,f-b\,g)\,(p+1)+(m\,(c\,e\,f+c\,d\,g-b\,e\,g)+e\,(p+1)\,(2\,c\,f-b\,g))\,x)\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  g*(d+e*x)^m*(a+b*x+c*x^2)^(p+1)/(c*(m+2*p+2)) +
  1/(c*(m+2*p+2))*Int[(d+e*x)^(m-1)*(a+b*x+c*x^2)^p*
    Simp[m*(c*d*f-a*e*g)+d*(2*c*f-b*g)*(p+1)+(m*(c*e*f+c*d*g-b*e*g)+e*(p+1)*(2*c*f-b*g))*x,x],x] /;
FreeQ[{a,b,c,d,e,f,g,p},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && GtQ[m,0] && NeQ[m+2*p+2,0] &&
  (IntegerQ[m] || IntegerQ[p] || IntegersQ[2*m,2*p]) && Not[IGtQ[m,0] && EqQ[f,0]]
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  g*(d+e*x)^m*(a+c*x^2)^(p+1)/(c*(m+2*p+2)) +
  1/(c*(m+2*p+2))*Int[(d+e*x)^(m-1)*(a+c*x^2)^p*
    Simp[c*d*f*(m+2*p+2)-a*e*g*m+c*(e*f*(m+2*p+2)+d*g*m)*x,x],x] /;
FreeQ[{a,c,d,e,f,g,p},x] && NeQ[c*d^2+a*e^2,0] && GtQ[m,0] && NeQ[m+2*p+2,0] &&
  (IntegerQ[m] || IntegerQ[p] || IntegersQ[2*m,2*p]) && Not[IGtQ[m,0] && EqQ[f,0]]
```

17: $\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\ \land\ c\,d^2-b\,d\,e+a\,e^2\neq 0\ \land\ m<-1$

Derivation: Quadratic recurrence 3b

Rule 1.2.1.3.17: If $b^2-4\,a\,c\neq 0\ \land\ c\,d^2-b\,d\,e+a\,e^2\neq 0\ \land\ m<-1$, then

$$\int (d+e\,x)^m\,(f+g\,x)\,\left(a+b\,x+c\,x^2\right)^p\,dx\ \longrightarrow$$

$$\frac{(e\,f-d\,g)\,(d+e\,x)^{m+1}\,\left(a+b\,x+c\,x^2\right)^{p+1}}{(m+1)\,\left(c\,d^2-b\,d\,e+a\,e^2\right)}+$$

$$\frac{1}{(m+1)\,\left(c\,d^2-b\,d\,e+a\,e^2\right)}\int (d+e\,x)^{m+1}\,\left(a+b\,x+c\,x^2\right)^p\,\left((c\,d\,f-f\,b\,e+a\,e\,g)\,(m+1)+b\,(d\,g-e\,f)\,(p+1)-c\,(e\,f-d\,g)\,(m+2\,p+3)\,x\right)\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  (e*f-d*g)*(d+e*x)^(m+1)*(a+b*x+c*x^2)^(p+1)/((m+1)*(c*d^2-b*d*e+a*e^2)) +
  1/((m+1)*(c*d^2-b*d*e+a*e^2))*Int[(d+e*x)^(m+1)*(a+b*x+c*x^2)^p*
    Simp[(c*d*f-f*b*e+a*e*g)*(m+1)+b*(d*g-e*f)*(p+1)-c*(e*f-d*g)*(m+2*p+3)*x,x],x] /;
FreeQ[{a,b,c,d,e,f,g,p},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && LtQ[m,-1] && (IntegerQ[m] || IntegerQ[p] || IntegersQ[2*m,2*p])
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  (e*f-d*g)*(d+e*x)^(m+1)*(a+c*x^2)^(p+1)/((m+1)*(c*d^2+a*e^2)) +
  1/((m+1)*(c*d^2+a*e^2))*Int[(d+e*x)^(m+1)*(a+c*x^2)^p*Simp[(c*d*f+a*e*g)*(m+1)-c*(e*f-d*g)*(m+2*p+3)*x,x],x] /;
FreeQ[{a,c,d,e,f,g,p},x] && NeQ[c*d^2+a*e^2,0] && LtQ[m,-1] && (IntegerQ[m] || IntegerQ[p] || IntegersQ[2*m,2*p])
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  (e*f-d*g)*(d+e*x)^(m+1)*(a+b*x+c*x^2)^(p+1)/((m+1)*(c*d^2-b*d*e+a*e^2)) +
  1/((m+1)*(c*d^2-b*d*e+a*e^2))*Int[(d+e*x)^(m+1)*(a+b*x+c*x^2)^p*
    Simp[(c*d*f-f*b*e+a*e*g)*(m+1)+b*(d*g-e*f)*(p+1)-c*(e*f-d*g)*(m+2*p+3)*x,x],x] /;
FreeQ[{a,b,c,d,e,f,g,m,p},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && ILtQ[Simplify[m+2*p+3],0] && NeQ[m,-1]
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  (e*f-d*g)*(d+e*x)^(m+1)*(a+c*x^2)^(p+1)/((m+1)*(c*d^2+a*e^2)) +
  1/((m+1)*(c*d^2+a*e^2))*Int[(d+e*x)^(m+1)*(a+c*x^2)^p*Simp[(c*d*f+a*e*g)*(m+1)-c*(e*f-d*g)*(m+2*p+3)*x,x],x] /;
FreeQ[{a,c,d,e,f,g,m,p},x] && NeQ[c*d^2+a*e^2,0] && ILtQ[Simplify[m+2*p+3],0] && NeQ[m,-1]
```

18: $\int \frac{f+g x}{(d+e x)\sqrt{a+b x+c x^2}}\,dx$ when $4 c (a-d)-(b-e)^2=0 \land f e (b-e)-2 g (b d-a e)=0 \land b d-a e\neq 0$

- Derivation: Integration by substitution

- Basis: If $4 c (a-d)-(b-e)^2=0 \land f e (b-e)-2 g (b d-a e)=0$, then $\frac{f+g x}{(d+e x)\sqrt{a+b x+c x^2}}=\frac{4 f (a-d)}{b d-a e}\,\text{Subst}\left[\frac{1}{4 (a-d)-x^2},x,\frac{2 (a-d)+(b-e) x}{\sqrt{a+b x+c x^2}}\right]\partial_x \frac{2 (a-d)+(b-e) x}{\sqrt{a+b x+c x^2}}$

- Rule 1.2.1.3.18: If $4 c (a-d)-(b-e)^2=0 \land f e (b-e)-2 g (b d-a e)=0 \land b d-a e\neq 0$, then

$$\int \frac{f+g x}{(d+e x)\sqrt{a+b x+c x^2}}\,dx \to \frac{4 f (a-d)}{b d-a e}\,\text{Subst}\left[\int \frac{1}{4 (a-d)-x^2}\,dx,x,\frac{2 (a-d)+(b-e) x}{\sqrt{a+b x+c x^2}}\right]$$

- Program code:

```
Int[(f_+g_.*x_)/((d_.+e_.*x_)*Sqrt[a_.+b_.*x_+c_.*x_^2]),x_Symbol] :=
  4*f*(a-d)/(b*d-a*e)*Subst[Int[1/(4*(a-d)-x^2),x],x,(2*(a-d)+(b-e)*x)/Sqrt[a+b*x+c*x^2]] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[4*c*(a-d)-(b-e)^2,0] && EqQ[e*f*(b-e)-2*g*(b*d-a*e),0] && NeQ[b*d-a*e,0]
```

19. $\int \frac{f+g\,x}{\sqrt{e\,x}\,\sqrt{a+b\,x+c\,x^2}}\,dx$ when $b^2-4\,a\,c\neq 0$

1: $\int \frac{f+g\,x}{\sqrt{x}\,\sqrt{a+b\,x+c\,x^2}}\,dx$ when $b^2-4\,a\,c\neq 0$

Derivation: Integration by substitution

Basis: $x^m\,F[x]=2\,\text{Subst}\left[x^{2\,m+1}\,F\left[x^2\right],\,x,\,\sqrt{x}\right]\,\partial_x\sqrt{x}$

Rule 1.2.1.3.19.1: If $b^2-4\,a\,c\neq 0$, then

$$\int \frac{f+g\,x}{\sqrt{x}\,\sqrt{a+b\,x+c\,x^2}}\,dx\to 2\,\text{Subst}\left[\int \frac{f+g\,x^2}{\sqrt{a+b\,x^2+c\,x^4}}\,dx,\,x,\,\sqrt{x}\right]$$

Program code:

```
Int[(f_+g_.*x_)/(Sqrt[x_]*Sqrt[a_+b_.*x_+c_.*x_^2]),x_Symbol] :=
  2*Subst[Int[(f+g*x^2)/Sqrt[a+b*x^2+c*x^4],x],x,Sqrt[x]] /;
FreeQ[{a,b,c,f,g},x] && NeQ[b^2-4*a*c,0]
```

```
Int[(f_+g_.*x_)/(Sqrt[x_]*Sqrt[a_+c_.*x_^2]),x_Symbol] :=
  2*Subst[Int[(f+g*x^2)/Sqrt[a+c*x^4],x],x,Sqrt[x]] /;
FreeQ[{a,c,f,g},x]
```

2: $\int \frac{f+g\,x}{\sqrt{e\,x}\,\sqrt{a+b\,x+c\,x^2}}\,dx$ when $b^2-4\,a\,c\neq 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{x}}{\sqrt{e\,x}}=0$

Rule 1.2.1.3.19.2: If $b^2-4\,a\,c\neq 0$, then

$$\int \frac{f+g x}{\sqrt{e x}\ \sqrt{a+b x+c x^2}}\,dx \to \frac{\sqrt{x}}{\sqrt{e x}}\int \frac{f+g x}{\sqrt{x}\ \sqrt{a+b x+c x^2}}\,dx$$

```
Int[(f_+g_.*x_)/(Sqrt[e_*x_]*Sqrt[a_+b_.*x_+c_.*x_^2]),x_Symbol] :=
  Sqrt[x]/Sqrt[e*x]*Int[(f+g*x)/(Sqrt[x]*Sqrt[a+b*x+c*x^2]),x] /;
FreeQ[{a,b,c,e,f,g},x] && NeQ[b^2-4*a*c,0]
```

```
Int[(f_+g_.*x_)/(Sqrt[e_*x_]*Sqrt[a_+c_.*x_^2]),x_Symbol] :=
  Sqrt[x]/Sqrt[e*x]*Int[(f+g*x)/(Sqrt[x]*Sqrt[a+c*x^2]),x] /;
FreeQ[{a,c,e,f,g},x]
```

20: $\int (d+e x)^m (f+g x) \left(a+b x+c x^2\right)^p\,dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0$

Derivation: Algebraic expansion

Basis: $f+g x == \frac{g (d+e x)}{e}+\frac{e f-d g}{e}$

Rule 1.2.1.3.20: If $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0$, then

$$\int (d+e x)^m (f+g x) \left(a+b x+c x^2\right)^p\,dx \to$$
$$\frac{g}{e}\int (d+e x)^{m+1} \left(a+b x+c x^2\right)^p\,dx+\frac{e f-d g}{e}\int (d+e x)^m \left(a+b x+c x^2\right)^p\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  g/e*Int[(d+e*x)^(m+1)*(a+b*x+c*x^2)^p,x] + (e*f-d*g)/e*Int[(d+e*x)^m*(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,f,g,m,p},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && Not[IGtQ[m,0]]
```

```
Int[(d_.+e_.*x_)^m_*(f_.+g_.*x_)*(a_+c_.*x_^2)^p_.,x_Symbol] :=
  g/e*Int[(d+e*x)^(m+1)*(a+c*x^2)^p,x] + (e*f-d*g)/e*Int[(d+e*x)^m*(a+c*x^2)^p,x] /;
FreeQ[{a,c,d,e,f,g,m,p},x] && NeQ[c*d^2+a*e^2,0] && Not[IGtQ[m,0]]
```